On the gravitational corrections in SGP4

Uncorrected draft: December 3, 2015

Martin Lara

Personal communication to Pablo Pita Leira

1 Introduction

Here, I provide a limited description of Vallado's implementation of the SGP4 program, making emphasis in the perturbation theory aspects, rather than in the efficient implementation of the formulas. My discussions restrict to `SUBROUTINE SGP4()`, of which I do not pay attention at all to the deep space part of the program, and even though I mention the drag corrections, this is just because they appear imbricated in the code, but I do not discuss anything about the drag perturbation theory.

```
* -----------------------------------------------------------------------------
*
*                              SUBROUTINE SGP4
*
*  this procedure is the sgp4 prediction model from space command. this is an
*
*  author        : david vallado                  719-573-2600   28 jun 2005
*------------------------------------------------------------------------------

      SUBROUTINE SGP4 ( whichconst, T, r, v, Error )
```

In the last section, numbered 6, I describe the modifications to be made for stating both the long- and short-period corrections in my proposed non-singular variables: See, my paper

> "Efficient formulation of the periodic corrections in Brouwer's gravity solution," *Mathematical Problems in Engineering*, Vol. 2015, Article ID 980652, 2015, 9 pp., doi:10.1155/2015/980652,

2 Preliminaries

Brouwer's theory relies on Delaunay variables

- $\ell = M$: mean anomaly
- $g = \omega$: argument of the periapsis
- $h = \Omega$: RAAN
- $L = \sqrt{\mu\, a}$: Delaunay action, where a is the semi-major axis
- $G = \sqrt{\mu\, p}$: total angular momentum, where $p = a(1 - e^2)$ is the orbit parameter (*semilatus rectum*), and e is the orbit eccentricity
- $H = G \cos I$: projection of the angular momentum on the earth's rotation axis, where I is the orbit inclination

However, Brouwer's theory finds trouble when evaluating the short-period corrections for the lower eccentricity orbits. The trouble is artificial, and is related to the singularity of Delaunay variables for circular orbits. Hence, SGP4 implements a different set of elements based on Lyddane's approach which completely avoids that trouble. In particular, the elements used in the computation of short-period corrections are[1]

$$F = \ell + g + h, \quad C = e \cos \omega, \quad S = e \sin \omega, \quad a, \quad I, \quad h$$

2.1 Initialization

The initialization of the code, provides the initial mean elements at epoch

$$I_0, \quad \Omega_0, \quad e_0, \quad \omega_0, \quad M_0, \quad n_0,$$

which should be obtained from TLEs. From them, Delaunay elements at epoch can be obtained as

$$\ell_0'' = M_0, \qquad g_0'' = \omega_0, \qquad h_0'' = \Omega_0,$$

$$L_0'' = (\mu^2/n_0)^{1/3}, \quad G_0'' = L_0''\sqrt{1 - e_0^2}, \quad H_0'' = G_0'' \cos I_0,$$

where the double prime notation is used for "secular" elements. Note that the Delaunay actions, viz. L, G, and H, are never computed in SGP4, which uses n, e, and I, instead.

[1]The original variables proposed by Lyddane were a, F, $e \cos \ell$, $e \sin \ell$, $\sin \frac{1}{2} I \cos h$, and $\sin \frac{1}{2} I \sin h$.

From these initial elements, Brouwer's theory provides the secular frequencies due to the earth's gravitational potential. Namely,

$$\dot{\ell}'' = \dot{\ell}''(n_0, e_0, I_0), \qquad \dot{g}'' = \dot{g}''(n_0, e_0, I_0), \qquad \dot{h}'' = \dot{h}''(n_0, e_0, I_0),$$

where the overdot means time derivative. These frequencies which remain constant for the evaluation of the theory at different dates.

Besides, the initialization process provides a series of coefficients needed to apply drag secular corrections as computed from Lane's theory.

2.2 Internal units

The code works with internal units of length LU (units of earth's radius $R_\oplus$ in km) and time TU (units of the orbit's period in min)

$$\mathrm{LU} = R_\oplus\,\mathrm{km}, \qquad \mathrm{TU} = 60\sqrt{(R_\oplus\,\mathrm{km})^3/(\mu\,\mathrm{km}^3/\mathrm{s}^2)}\,\mathrm{min}$$

where μ is the earth's gravitational constant; $\mu = 1\,\mathrm{LU}^3/\mathrm{TU}^2$ in internal units. Besides, the expansion of the gravitational potential in SGP4 is limited to the (unnormalized) zonal harmonics J_2, J_3, and J_4.

```
*-----------------------------------------------------------------------------
        CALL getgravconst( whichconst, tumin, mu, radiusearthkm, xke,
     &       j2, j3, j4, j3oj2 )
*-----------------------------------------------------------------------------
```

This selection of units helps in optimizing the code, but it is irrelevant for the description of the algorithms. Therefore, following descriptions include explicitly both the earth's gravitational parameter as well as the earth's equatorial radius.

3 Secular terms

The next step is to update the secular elements at epoch to the desired date given by the time t.

3.1 Gravity corrections

Brouwer's gravitational corrections are applied first

$$\ell''_j = M_0 + \dot{\ell}''\,t, \quad g''_j = \omega_0 + \dot{g}''\,t, \quad h''_j = \Omega_0 + \dot{h}''\,t,$$

where the subindex j stands for gravitational effects only.

3.2 Drag secular and long-period corrections

Next, the secular corrections due to the atmospheric drag are incorporated; in particular

$$\begin{aligned}
\delta_h &= (21/4) J_2 (R_\oplus^2/p_0^2) \beta_0^2 \cos I_0 \, (n_0 C_1 t^2) \quad \texttt{units?} \\
\delta_L &= (L''/L_0) = 1 - C_1 t - D_2 t^2 - D_3 t^3 - D_4 t^4 \\
\delta_e &= (e_0'' - e'') = B^* C_4 t \\
\delta_\ell &= (\ell'' - \ell_j'')/n_0'' = T_2\, t^2 + T_3\, t^3 + T_4\, t^4 + T_5\, t^5
\end{aligned}$$

where $\beta_0^2 = 1 - e_0^2$, and $p = a_0 \beta_0^2$. The necessary constants and coefficients C_k, D_k, T_k, B^*, etc. have been evaluated at the initialization stage.

Remind that I do not deal with the deep space secular corrections

Then compute the secular elements (not exactly secular: they mix long-period terms from drag)

$$\begin{aligned}
\ell'' &= \ell_j'' + n_0'' \delta_\ell + \Delta_\ell \\
g'' &= g_j'' - \Delta_\ell \\
h'' &= h_j'' + \delta_h \\
a'' &= L''^2/\mu = (L_0^2/\mu)\,(L''/L_0)^2 = a_0\, \delta_L^2 = (\mu/n_0^2)^{1/3} \delta_L^2 \\
e'' &= e_0'' - \delta_e - B^* C_5 \left[\sin(\ell_j'' + \Delta_\ell) - \sin \ell_0'' \right] \\
I'' &= I_0''.
\end{aligned}$$

where Δ_ℓ is a long-period correction (see `SPACETRAK rep. #3, p. 41`).

The secular mean motion $n'' = (\mu/a''^3)^{1/2}$ is also computed.

4 Geopotential long-period corrections

Remind that I do not deal with the deep space secular corrections

Brouwer long-period gravitational corrections are reformulated in Lydane's. At the precision of SGP4, there are only corrections for F and S. Then,

$$\begin{aligned}
F' &= (\ell'' + g'' + h'') - \frac{J_3}{4J_2} \frac{R_\oplus}{p''} C'' \frac{3 + 5\cos I''}{1 + \cos I''} \sin I'' \\
S' &= e'' \sin g'' - \frac{J_3}{2J_2} \frac{R_\oplus}{p''} \sin I'' \\
C' &= C'' \\
a' &= a''
\end{aligned}$$

$$
\begin{aligned}
h' &= h'' \\
I' &= I''.
\end{aligned}
$$

where $C'' = e'' \cos g''$, $p'' = a''(1 - e''^2)$. The single prime notation means that Lyddane elements include long-period effects. Note that $n' = n''$.

```
* ------------------------ LONG PERIOD PERIODICS ----------------------
      IF(METHOD .EQ. 'd') THEN
          SINIP =  DSIN(XINCP)
          COSIP =  DCOS(XINCP)
          AYCOF = -0.5D0*J3OJ2*SINIP ! -(1/2)*(J3/J2)*SIN(I")
c         sgp4fix for divide by zero with xincp = 180 deg
          if (dabs(cosip+1.0).gt. 1.5d-12) THEN
! -(1/4)*(J3/J2)*SIN(I")*(3+5*COS(I"))/(1+COS(I"))
              XLCOF  = -0.25D0*J3OJ2*SINIP*
     &                  (3.0D0+5.0D0*COSIP)/(1.0D0+COSIP)
            else
              XLCOF  = -0.25D0*J3OJ2*SINIP*
     &                  (3.0D0+5.0D0*COSIP)/temp4
            ENDIF
        ENDIF
      AXNL = Eccp*DCOS(Argpp)
!        C'= e"*COS(g"): no hay correccion de largo periodo
      TEMP = 1.0D0 / (AM*(1.0D0-Eccp*Eccp)) ! 1/p" = 1/(a"*y"^2)  y=eta
      AYNL = Eccp*DSIN(Argpp) + TEMP*AYCOF
!        S'= e"*SIN(g") + (1/p")*(-(1/2)*(J3/J2)*SIN(I"))
      XL   = Mp + Argpp + nodep + TEMP*XLCOF*AXNL
!        F'= (M"+g"+h")
!           +(1/p")(-1/4)*(J3/J2)*SIN(I")*e"*COS(g")*(3+5*c")/(1+c")
!        L'= L"
!        I'= I"
!        h'= h"
```

4.1 Solve Kepler equation

Solve the Kepler equation

$$U = \Psi + S' \cos \Psi - C' \sin \Psi,$$

where $U = F' - h'$, to compute the anomaly $\Psi = E' + g'$, where E is the eccentric anomaly. The Newton-Raphson iterations start from $\Psi_0 = U$.

```
* ------------------------- SOLVE KEPLER'S EQUATION -------------------
! (F'-h") = (E"+g") + S'*COS(E"+g") - C'*SIN(E"+g"): comprobar
! U = Y + S'*COS(Y) - C'*SIN(Y)
! fun(Y) = U - Y - S'*COS(Y) + C'*SIN(Y) = 0;
! fun(Y0+(Y-Y0))= fun(Y0)+(Y-Y0)*dfu(Y0) = 0;
!                 dfu = D[fun,Y] = -1 + S'*SIN(Y) + C'*COS(Y)
! Y1= Y0 - fun(Y0)/dfu(Y0);
! Y(n+1) = Yn - fun(Yn)/dfu(Yn)
      U    = DMOD(XL-nodep,TwoPi) ! U = F'-h" = M"+g"; Y = E"+g"
      EO1  = U ! Y0 = U
      ITER=0
c   sgp4fix for kepler iteration
c   the following iteration needs better limits on corrections
      Temp = 9999.9D0
      DO WHILE ((Temp.ge.1.0D-12).and.(ITER.lt.10))
          ITER=ITER+1
          SINEO1= DSIN(EO1)
          COSEO1= DCOS(EO1)
          TEM5  = 1.0D0 - COSEO1*AXNL - SINEO1*AYNL
! -dfu = 1 - S'*SIN(Y) - C'*COS(Y)
          TEM5  = (U - AYNL*COSEO1 + AXNL*SINEO1 - EO1) / TEM5
! U - S'*COS(Y)+ C'*SIN(Y)
          Temp  = DABS(Tem5)
          IF(Temp.gt.1.0D0) Tem5=Tem5/Temp ! Stop excessive correction
          EO1   = EO1+TEM5
        ENDDO
```

4.2 Compute polar variables

Change from Lyddane non-singular variables to polar-nodal variables

$$(r, \theta, R, \Theta) \longrightarrow (F, C, S, a),$$

To do that, first compute g' and e' from

$$C' = e' \cos g', \qquad S' = e' \sin g'$$

then, compute $E' = \Psi - g'$, $L' = \sqrt{\mu a'}$, and $\beta' = \sqrt{1 - e'^2}$. It follows, the usual transformation

$$r' = a(1 - e' \cos E')$$

$$
\begin{aligned}
R' &= (L'/r')e' \sin E' \\
\Theta' &= L'\beta' \\
\sin f' &= (a'/r')\beta' \sin E', \qquad \cos f' = (a'/r')(\cos E' - e'), \\
\theta' &= f' + g'
\end{aligned}
$$

Note: SGP4 uses $r\dot{\theta} = \Theta/r$ instead of Θ

```
* ----------------- SHORT PERIOD PRELIMINARY QUANTITIES ---------------
      ECOSE = AXNL*COSEO1+AYNL*SINEO1 ! e'*COS(E')
      ESINE = AXNL*SINEO1-AYNL*COSEO1 ! e'SIN(E')
      EL2   = AXNL*AXNL+AYNL*AYNL ! e'**2
      PL    = AM*(1.0D0-EL2) ! p=a*(1-e**2); como mu=1, p=Z**2
c     semi-latus rectum < 0.0
      IF ( PL .lt. 0.0D0 ) THEN
          Error = 4
        ELSE
          RL    = AM*(1.0D0-ECOSE) ! r' = a'*(1-e'*COS(E'))
          RDOTL = DSQRT(AM)*ESINE/RL ! R'
          RVDOTL= DSQRT(PL)/RL ! (Z'/r')
          BETAL = DSQRT(1.0D0-EL2) ! y'
          TEMP  = ESINE/(1.0D0+BETAL)
          SINU  = AM/RL*(SINEO1-AYNL-AXNL*TEMP) ! SIN(z')
          COSU  = AM/RL*(COSEO1-AXNL+AYNL*TEMP) ! COS(z')
          SU    = DATAN2(SINU,COSU) ! z'
          SIN2U = (COSU+COSU)*SINU ! SIN(2z')
          COS2U = 1.0D0-2.0D0*SINU*SINU ! COS(2z')
          TEMP  = 1.0D0/PL ! 1/p'
          TEMP1 = 0.5D0*J2*TEMP ! (1/2)*(J2/p)
          TEMP2 = TEMP1*TEMP ! (1/2)*(J2/p**2)
```

5 Compute short-period gravitational corrections

Compute $p' = a'\beta'^2$, $c' = \cos I'$, $s' = \sin I'$, and $\epsilon = \frac{1}{2}J_2(R_\oplus/p')^2$. Then,

$$
\begin{aligned}
r - r' &= r'\epsilon\left[\frac{p'}{r'}\frac{1}{2}(1-c'^2)\cos 2\theta' - \frac{3}{2}\beta'(3c'^2-1)\right] \\
\theta - \theta' &= -\epsilon\frac{1}{4}(7c'^2-1)\sin 2\theta' \\
h - h' &= \frac{3}{2}\epsilon c' \cos\theta'
\end{aligned}
$$

$$
\begin{aligned}
R - R' &= -p' n' \epsilon (1 - c'^2) \sin 2\theta' \\
I - I' &= \frac{3}{2} \epsilon c'\, s' \cos 2\theta' \\
\frac{\Theta}{r} - \frac{\Theta'}{r'} &= \frac{\Theta'}{r'} \epsilon \frac{r'}{p'} \left[\left(1 - c'^2\right) \cos 2\theta' + \frac{3}{2} \left(3c'^2 - 1\right) \right]
\end{aligned}
$$

```
* ------------------ UPDATE FOR SHORT PERIOD PERIODICS ----------------
          IF(METHOD .EQ. 'd') THEN
              COSISQ = COSIP*COSIP
              CON41  = 3.0D0*COSISQ - 1.0D0
              X1MTH2 = 1.0D0 - COSISQ
              X7THM1 = 7.0D0*COSISQ - 1.0D0
            ENDIF
!
! Basically, corrections to polar-nodal variables
!
          mr   = RL*(1.0D0 - 1.5D0*TEMP2*BETAL*CON41) +
     &           0.5D0*TEMP1*X1MTH2*COS2U
!            r = r' - (r'/p')*(3/2)*(1/2)*(J2/p')*y'*(3c'**2-1)
!                   + (1/2)*(1/2)*(J2/p')*(1-c'**2)*COS(2z')
          SU   = SU - 0.25D0*TEMP2*X7THM1*SIN2U
!            z = z'-(1/4)*(1/2)*(J2/p'**2)*(7c'**2-1)*SIN(2z')
          XNODE= nodep + 1.5D0*TEMP2*COSIP*SIN2U
!            h = h' + (3/2)*(1/2)*(J2/p**2)*c*SIN(2z')
          XINC = XINCP + 1.5D0*TEMP2*COSIP*SINIP*COS2U
!            I = I' + (3/2)*(1/2)*(J2/p**2)*c*s*COS(2z')
          mv   = RDOTL - XN*TEMP1*X1MTH2*SIN2U / XKE
!            R = R' - n*(1/2)*(J2/p')*(1-c'**2)*SIN(2z')
!            R = R' - (Z/p**2)*(1/4)*(J2/p)*(2*(1-c**2)*SIN(2z') + e*(...))
!                     (Z/p^2)=((mu*p)^(1/2)/p^2)=mu^(1/2)/p^(3/2)
!                            =mu**(1/2)/a**(3/2)*(1+(3/2)*e**2+...)
          RVDOT= RVDOTL + XN*TEMP1* (X1MTH2*COS2U+1.5D0*CON41) / XKE
!         (Z/r) = (Z'/r') + n*(1/2)*(J2/p)*((1-c'**2)*COS(2z') + (3/2)*(3*c'**2-1))
!            Z = Z' - (3/2)*(Z'/p')*(1/2)*(J2/p')*(1-c'**2)*COS(2z')
```

Finally, the state vector is computed from the standard transformation from Cartesian to polar-nodal variables

$$
\begin{pmatrix} x & \dot{x} \\ y & \dot{y} \\ z & \dot{z} \end{pmatrix} = R_3(-h)\, R_1(-I)\, R_3(-\theta) \begin{pmatrix} r & \dot{r} = R \\ 0 & r\dot{\theta} = \Theta/r \\ 0 & 0 \end{pmatrix} \tag{1}
$$

where R_1 and R_3 are the usual rotation matrices about the x and z axes, respectively

```
* ------------------------- ORIENTATION VECTORS -----------------------
! standard transformation polar-nodal to Cartesian
!
          SINSU=  DSIN(SU)
          COSSU=  DCOS(SU)
          SNOD =  DSIN(XNODE)
          CNOD =  DCOS(XNODE)
          SINI =  DSIN(XINC)
          COSI =  DCOS(XINC)
          XMX  = -SNOD*COSI
          XMY  =  CNOD*COSI
          UX   =  XMX*SINSU + CNOD*COSSU
          UY   =  XMY*SINSU + SNOD*COSSU
          UZ   =  SINI*SINSU
          VX   =  XMX*COSSU - CNOD*SINSU
          VY   =  XMY*COSSU - SNOD*SINSU
          VZ   =  SINI*COSSU

* ----------------------- POSITION AND VELOCITY -----------------------
          r(1) = mr*UX * RadiusEarthkm
          r(2) = mr*UY * RadiusEarthkm
          r(3) = mr*UZ * RadiusEarthkm
          v(1) = (mv*UX + RVDOT*VX) * VKmPerSec
          v(2) = (mv*UY + RVDOT*VY) * VKmPerSec
          v(3) = (mv*UZ + RVDOT*VZ) * VKmPerSec
        ENDIF
```

6 My approach

Start from section 4 as follows.

- solve Kepler equation to compute the eccentric anomaly, and then the true anomaly

$$\ell'' = E'' - e'' \sin E'', \qquad \tan\frac{f''}{2} = \sqrt{\frac{1+e''}{1-e''}}\tan\frac{E''}{2},$$

- transform the Delaunay variables to my non-singular variables

- 1st, Delaunay to polar-nodal: $(\ell, g, h, L, G, H) \longrightarrow (r, \theta, \nu, R, \Theta, N)$
- then, $\psi = \theta + \nu$, $\xi = \sin I \sin\theta$, $\chi = \sin I \cos\theta$, r, R, Θ

- apply my long-period corrections (simplified according SGP4 criteria)

$$\begin{aligned}
\delta\psi &= 2\epsilon_3\,\chi \\
\delta\xi &= \chi\,\delta\psi \\
\delta\chi &= -\xi\,\delta\psi \\
\delta r &= p\epsilon_3\,\xi, \\
\delta R &= (\Theta/r)\,\epsilon_3\,(p/r)\,\chi \\
\delta\Theta &= \Theta\epsilon_3\left[\left(\frac{p}{r}-1\right)\xi - \frac{pR}{\Theta}\chi\right].
\end{aligned}$$

where, $\epsilon_3 = \frac{1}{2}(J_3/J_2)(\alpha/p)$

- apply my short-period corrections (simplified according SGP4 criteria)

$$\begin{aligned}
\Delta\psi &= -\epsilon_2\frac{1+7c}{1+c}\,\xi\,\chi, \\
\Delta\xi &= -\epsilon_2\left(\chi^2 - 3c^2\right)\xi, \\
\Delta\chi &= \epsilon_2\left(\xi^2 - 3c^2\right)\chi, \\
\Delta r &= \epsilon_2 r\left[(\xi^2-\chi^2) - 3(1-3c^2)\right], \\
\Delta R &= 4\epsilon_2\,(\Theta/r)\xi\,\chi, \\
\Delta\Theta &= 3\epsilon_2\,\Theta(\xi^2-\chi^2),
\end{aligned}$$

where $c = \cos I$, $s = \sin I$, and $\epsilon_2 = -\frac{1}{4}(\alpha/p)^2 J_2$

- compute the state vector

$$\begin{aligned}
x &= r\,(b\cos\psi + q\sin\psi), \\
y &= r\,(b\sin\psi - q\cos\psi), \\
z &= r\,\xi, \\
X &= x\,(R/r) - (\Theta/r)\,(q\cos\psi + \tau\sin\psi), \\
Y &= y\,(R/r) - (\Theta/r)\,(q\sin\psi - \tau\cos\psi), \\
Z &= z\,(R/r) + (\Theta/r)\,\chi,
\end{aligned}$$

where

$$b = 1 - \frac{\xi^2}{1+c}, \qquad \tau = 1 - \frac{\chi^2}{1+c}, \qquad q = \frac{\xi\,\chi}{1+c},$$

and $c = N/\Theta$. Remark that N is an integral of the zonal problem and, therefore, its value is always known from given initial conditions $N = H_0''$.